# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Contributor: Jameson Pugh <imntreal@gmail.com>

pkgname=ell
_commit=726ca36c2a494d7b22222f35a0cedf04114cdabf  # tags/0.5
pkgver=0.5
pkgrel=1
pkgdesc="Embedded Linux library"
arch=(x86_64)
url="https://git.kernel.org/pub/scm/libs/ell/ell.git/"
license=('LGPL2.1')
depends=('glibc')
makedepends=('git')
source=("git+https://git.kernel.org/pub/scm/libs/ell/ell.git/#commit=$_commit")
sha256sums=('SKIP')

prepare() {
  cd "$pkgname"
  ./bootstrap
}

build() {
  cd "$pkgname"
  ./configure --prefix=/usr
  make
}

check() {
  cd "$pkgname"
  make -k check
}

package() {
  cd "$pkgname"
  make DESTDIR="$pkgdir/" install
}
